

SNAKE GAME

高木 真一郎

★はじめに★

ラジカセ感覚で使えるホーム/パソコンMSX規格のマシンもちらほらと市場にでてきた。さっそくゲームで楽しんでみたいという人のためにプログラムをつくってみた。

これは画面上をはいまれるへびを動かして次々にあらわれるエサを食べるゲームである。比較的短めだがMSX-BASIC特有のDRAW文(図形を画面に描く命令)やサウンド機能を生かしたものにしている。

★プログラム入力の仕方★

まず電源を入れ、BASICが起動するのを確認。キーボード左上の“F2”のキーを押す。これには、

“AUTO”コマンドが定義されていて、行番号(プログラムの左側の数字)を10きざみに自動的に出力する。プログラムをリスト通りに入力する。1行入れ終わることに必ず、“リターン”キーを押すのを忘れずに。最後まで入れたら“CTRL”キーを押しながら、“STOP”キーを押す。これで、AUTO状態からもとの状態に復帰、入力完了だ。

プログラムを“RUN”する前に、テープに“CSAVE”しておこう。カセットテープを録音状態にして、csave “snake” と入力した後リターンキーを押す。“OK”の文字が出たらCSAVE完了。“F5”のキー

を押すとゲームがスタートする。

もしも途中でゲームが止まってしまったら、エラーメッセージに注目。LIST行番号リターンで、エラーのあった行を出力。Syntax error ならスベルミス、Illegal function call なら、数字が間違っているはずだ。リストとにらめっこして直してくれ。エラーが全部なくなったら、もう一度csaveし直すことを忘れずに。

★遊び方★

まずRUN[RET]でスタートすると、画面にへびとエサがあらわれる。へびを動かすにはカーソル移動キー[←]を使う。へびは[←]キーで左に、

[→]キーで右に動かすことができるが、この場合の左右はへびの進行方向にむかっているものであることに注意すること。こんなふうにへびを左右に動かしてエサを食べる。

エサは初めは1つだけあらわれるが一定時間内に食べないと3個増える(時間は画面下に帯で表示される)。エサを1個食べるごとに胴体は長くなっていく。


エサを全部食べると画面上中央に出口ができ、そこをぬけると次の面に進むことができるのである。パターンは、6面まであり、パターンが進むにつれ当然難しくなってくるのでご注意! 外側のカベや自分の胴体

に衝突すると、へびは死んでしまうのだ。3匹のへびが死ぬとゲームオーバー。

★プログラムの改良★

ジョイスティックを使うときは、140行の“ST=0”を“ST=1”に変える。また、へびを増やすときは140行の“SN=3”、エサの数を変えるときは1160行の“LF=10*LV”の定数をそれぞれ変えることが必要。面の数を増やしたい場合は、1150行・1380行を変更し、(面数はSC)プログラム末尾に壁を描くサブルーチンを作ればOK。

```
10 *****
20 *
30 * SNAKE GAME *
40 *
50 * FOR MSX BASIC *
60 *
70 * BY S . T *
80 *****
90
100 SCREEN 2,0,0:COLOR ,0,0:DEFINT A-Z:A=RND(-TIME)
110 DIM X(500),Y(500)
120 GOSUB 1390:'キャラクタデータ
130 OPEN"GRP:"AS#1
140 SC=1:SN=3:PO=0:LV=1:ST=0
150 GOSUB 1050:'シヨキゲーム
160
170 '=== シヨキセッテイ ===
180
190 HD=0:LS=0:X(HD)=16:Y(HD)=22:DL=5:DR=0:SF=1:KS=0:TI=255:SD=0
200 PSET(128,176),0:COLOR 8:DRAW HD$
210 FX=INT(RND(1)*28)+1:FY=INT(RND(1)*20)+2
220 IF POINT(FX*8+3,FY*8+3) THEN 210
230 PSET(FX*8,FY*8),0:COLOR 14:DRAW FD$
240
250 '=== メインルーチン ===
260
270 IF LF=0 THEN 400
280 TI=TI-4:LINE(TI,187)-(TI+4,191),13,BF
290 IF TI>0 THEN 400
300 SF=SF+3:LF=LF+3:TI=255:LINE(0,186)-(255,191),2,BF
310 FOR N=1 TO 3
320 FX=INT(RND(1)*30+1)
330 FY=INT(RND(1)*20+2)
340 IF POINT(FX*8+4,FY*8+3)<>0 THEN 320
350 PSET(FX*8,FY*8),0:COLOR 14:DRAW FD$
360 NEXT
370
380 LINE(224,0)-(239,7),0,BF:COLOR 15:PSET(224,0),0:PRINT #1,USING"###";LF
390
400 KD=STICK(ST)
410 IF KD=KS THEN 430
420 DR=DR-(KD=3)-(KD=7)*3:DR=DRMOD 4:KS=KD
430 SD=1-SD
440 IF SD THEN PLAY"O6L32E" ELSE PLAY"O6L32C"
450 HX=X(HD):HY=Y(HD)
460 NX=HX-(DR=1)+(DR=3)
470 NY=HY-(DR=2)+(DR=0)
480 PSET(HX*8,HY*8),0:COLOR 0:DRAW HD$
490 PSET(HX*8,HY*8),0:COLOR 5:DRAW BD$
500 PN=POINT(NX*8+4,NY*8+3)
510 IF PN THEN 620
520 PSET(NX*8,NY*8),0:COLOR 8:DRAW HD$
530 HD=HD+1
540 IF HD=501 THEN HD=0
550 X(HD)=NX:Y(HD)=NY
560 IF DL THEN DL=DL-1:GOTO 600
570 PSET(X(LS)*8,Y(LS)*8),0:COLOR 0:DRAW BD$
580 LS=LS+1
590 IF LS=501 THEN LS=0
600 IF NX=16 AND NY=1 THEN 980 ELSE 250
```



```

610 '
620 '=== ハンテイ ===
630 '
640 IF PN<>14 THEN 840
650 PLAY"O6L64C#", "O6L64D", "O6L64G#"
660 LINE(0,186)-(255,191),2,BF:TI=255
670 PO=PO+SC*10
680 PSET(NX*8,NY*8),0:COLOR0:DRAW FD$
690 DL=DL+8:PSET(NX*8,NY*8),0:COLOR 0:DRAW HD$
700 LF=LF-1
710 IF LF=0 THEN LINE(120,8)-(143,15),8,BF:LINE(128,8)-(135,15),0,BF
720 LINE(224,0)-(239,7),0,BF
730 COLOR 15:PSET(224,0),0:PRINT #1,USING"###";LF
740 SF=SF-1
750 IF SF THEN 520
760 IF LF=0 THEN 810
770 SF=1
780 FX=INT(RND(1)*30)+1:FY=INT(RND(1)*20)+2
790 IF POINT(FX*8+4,FY*8+3)<>0 THEN 780
800 PSET(FX*8,FY*8),0:COLOR 14:DRAW FD$
810 LINE(64,0)-(103,7),0,BF:PSET(64,0),0:PRINT #1,USING"#####";PO
820 GOTO 520
830 '
840 ' === GAME OVER ===
850 '
860 PLAY"O7T250L3CR64CR64L6CL6R64L2CR64L3D#R64L8DR64R64L3DL6CR64L3CR64O6L5B07L2C"
870 LINE( NX*8+3,NY*8+3)-(INT(RND(1)*256),INT(RND(1)*192)),INT(RND(1)*16)
880 IF PLAY(0) THEN 870
890 SN=SN-1
900 IF SN THEN 150
910 LINE(72,52)-(151,75),1,BF
920 PSET(80,60),1:PRINT#1,"GAME OVER"
930 LINE(48,8)-(153,15),0,BF
940 PSET(50,8),0:COLOR 5:PRINT#1,"<- TO RE-PLAY"
950 IF STICK(ST)<>7 THEN 950
960 GOTO140
970 '
980 '=== 1 メン クリア ===
990 '
1000 PLAY"O7L32C2DEFGABL408C07"
1010 SC=SC+1:GOTO 150
1020 '
1030 '=== メン ノ セツテイ ===
1040 '
1050 CLS
1060 COLOR 2
1070 LINE(6,14)-(250,186),,B
1080 PAINT(0,0),2,2
1090 LINE(16,0)-(103,7),0,BF
1100 COLOR15:PSET(16,0),1:PRINT #1,USING"Score:#####";PO
1110 LINE(120,0)-(175,7),0,BF
1120 COLOR15:PSET(120,0),1:PRINT #1,USING"Snake:#";SN
1130 LINE(184,0)-(239,7),0,BF
1140 '
1150 ON SC-1 GOSUB 1180,1200,1230,1280,1320,1380
1160 LF=10*LV:COLOR 15:PSET(184,0),0:PRINT #1,"Food:";USING"###";LF
1170 RETURN
1180 LINE(56,96)-(175,103),2,BF
1190 RETURN
1200 LINE(56,96)-(175,103),2,BF
1210 LINE(120,58)-(127,143),2,BF
1220 RETURN
1230 LINE(80,64)-(159,71),2,BF
1240 LINE(80,96)-(159,103),2,BF
1250 LINE(80,128)-(159,135),2,BF
1260 LINE(80,64)-(87,135),2,BF
1270 RETURN
1280 LINE(56,56)-(199,143),2,BF
1290 LINE(64,64)-(191,136),0,BF
1300 LINE(120,56)-(135,63),0,BF
1310 RETURN
1320 LINE(56,56)-(199,143),2,BF
1330 LINE(120,56)-(135,143),0,BF
1340 LINE(64,64)-(112,136),0,BF
1350 LINE(144,64)-(191,136),0,BF
1360 LINE(70,96)-(150,111),0,BF
1370 RETURN
1380 SC=1:LV=LV+1:RETURN
1390 ' === キャラクタ テイキ` ===
1400 HD$="BR2R4F1D4G1L4H1U4E1"
1410 BD$="BD1BR4D3U1L1R2"
1420 FD$="BR3R2F1L4G1R5D1L5F1R3G1"
1430 RETURN

```